Rozklad mnohočlenu na součin ireducibilních činitelů

```
> restart;
```

PŘÍKLAD 1: Polynom $f(x) = x^8 + x^4 + 1$ **graficky znázorněte a proveďte jeho rozklady v různých oborech integrity**

```
> f:=x^8+x^4+1;
```

$$f := x^8 + x^4 + 1$$

```
> plot(f,x=-1/2..1/2);
```



Příkaz **factor**, jehož použití se nabízí, ne vždy provede rozklad v symbolickém režimu. Jak vidíme dále, výsledkem rozkladu nad tělesem reálných či komplexních čísel jsou činitele, jejichž koeficienty jsou vyjádřeny jenom přibližně.

```
> Rozklad_Q:=factor(f);
```

$$Rozklad\_Q := (x^2 - x + 1)\,(x^2 + x + 1)\,(x^4 - x^2 + 1)$$

```
> factor(f,real);
```

$$(x^2 + 1.732050808\,x + 1.000000000)\,(x^2 + 1.000000000\,x + 1.000000000)$$
$$(x^2 - 1.000000000\,x + 1.000000000)\,(x^2 - 1.732050808\,x + 1.000000000)$$

```
> factor(f,complex);
```

$$(x + 0.8660254038 + 0.5000000000\,I)\,(x + 0.8660254038 - 0.5000000000\,I)$$
$$(x + 0.5000000000 + 0.8660254038\,I)\,(x + 0.5000000000 - 0.8660254038\,I)$$
$$(x - 0.5000000000 + 0.8660254038\,I)\,(x - 0.5000000000 - 0.8660254038\,I)$$
$$(x - 0.8660254038 + 0.5000000000\,I)\,(x - 0.8660254038 - 0.5000000000\,I)$$

Pokud chceme rozklad v R[x] v uzavřeném tvaru, je třeba druhým parametrem příkazu factor specifikovat těleso koeficientů ireducibilních činitelů.

```
> Rozklad_R:=factor(f,sqrt(3));
```

$$Rozklad\_R := (x^2 - x + 1)\,(x^2 + x + 1)\,(x^2 + x\sqrt{3} + 1)\,(x^2 - x\sqrt{3} + 1)$$

Symbolického vyjádření ireducibilních činitelů a úplného rozkladu na lineární činitele v C[x] dosáhneme následující kombinací příkazů **polytools[split]** a **convert**.

```
> Roz_f:=polytools[split](f,x);
```

$$Roz\_f := (x + \mathrm{RootOf}(\_Z^2 + \_Z + 1))\,(x + \mathrm{RootOf}(\_Z^2 + \mathrm{RootOf}(\_Z^2 + \_Z + 1)))\,(x$$
$$+ \mathrm{RootOf}(\_Z^2 + \mathrm{RootOf}(\_Z^2 + \_Z + 1))$$

$$+\mathrm{RootOf}(\_Z^2+\_Z+1)\,\mathrm{RootOf}(\_Z^2+\mathrm{RootOf}(\_Z^2+\_Z+1)))$$
$$(x-\mathrm{RootOf}(\_Z^2+\mathrm{RootOf}(\_Z^2+\_Z+1)))\,(x-1-\mathrm{RootOf}(\_Z^2+\_Z+1))\,(x$$
$$-\mathrm{RootOf}(\_Z^2+\mathrm{RootOf}(\_Z^2+\_Z+1))$$
$$-\mathrm{RootOf}(\_Z^2+\_Z+1)\,\mathrm{RootOf}(\_Z^2+\mathrm{RootOf}(\_Z^2+\_Z+1)))$$
$$(x+1+\mathrm{RootOf}(\_Z^2+\_Z+1))\,(x-\mathrm{RootOf}(\_Z^2+\_Z+1))$$

```
> Rozklad_C:=convert(Roz_f,radical);
```

$$Rozklad\_C := \left(x-\frac{1}{2}+\frac{1}{2}I\sqrt{3}\right)\left(x+\sqrt{\frac{1}{2}-\frac{1}{2}I\sqrt{3}}\right)$$
$$\left(x+\sqrt{\frac{1}{2}-\frac{1}{2}I\sqrt{3}}+\left(-\frac{1}{2}+\frac{1}{2}I\sqrt{3}\right)\sqrt{\frac{1}{2}-\frac{1}{2}I\sqrt{3}}\right)\left(x-\sqrt{\frac{1}{2}-\frac{1}{2}I\sqrt{3}}\right)\left(x-\frac{1}{2}-\frac{1}{2}I\sqrt{3}\right)$$
$$\left(x-\sqrt{\frac{1}{2}-\frac{1}{2}I\sqrt{3}}-\left(-\frac{1}{2}+\frac{1}{2}I\sqrt{3}\right)\sqrt{\frac{1}{2}-\frac{1}{2}I\sqrt{3}}\right)\left(x+\frac{1}{2}+\frac{1}{2}I\sqrt{3}\right)\left(x+\frac{1}{2}-\frac{1}{2}I\sqrt{3}\right)$$

Opakem uvedených příkazů pro rozklad na součin činitelů je příkaz **expand** pro roznásobení součinu (Jeho výsledek ještě zjednodušíme příkazem **simplify**).

```
> simplify(expand(Rozklad_C));
```

$$x^8+x^4+1$$

PŘÍKLAD 2: Rozložte na součin ireducibilních činitelů v Q[x], R[x] a C[x] mnohočlen $x^4-x^3-x^2-x-2$.

```
> restart;
> P:=x^4-x^3-x^2-x-2;
```

$$P := x^4-x^3-x^2-x-2$$

Můžeme použít příkaz **factor**:

Rozklad v Q[x]

```
> factor(P);
```

$$(x-2)\,(x+1)\,(x^2+1)$$

Rozklad v R[x]:

```
> factor(P,real);
```

$$(x+1.)\,(x-2.)\,(x^2+1.)$$

Rozklad v C[x]

```
> factor(P,complex);
```

$$(x+1.)\,(x+1.\,I)\,(x-1.\,I)\,(x-2.)$$

Chceme-li se vyhnout zápisu koeficeintů výsledných kořenových činitelů desetinnými čísly, použijeme raději příkaz **split** z balíčku příkazů **polytools**, který kombinujeme s příkazem **convert**.

```
> Rozklad_P:=polytools[split](P,x);
```

$$Rozklad\_P := (x+1)(x-2)(x - \mathrm{RootOf}(\_Z^2+1))(x + \mathrm{RootOf}(\_Z^2+1))$$

```
> convert(Rozklad_P,radical);
```

$$(x+1)(x-2)(x-I)(x+I)$$

Grafické znázornění:

```
> plot(P,x=-3..3,y=-5..5);
```



PŘÍKLAD 3: Rozložte na součin ireducibilních činitelů v R[x] a C[x] mnohočlen $x^4 - x^2 - 2$.

```
> restart;
> Q:=x^4-x^2-2;
```

$$Q := x^4 - x^2 - 2$$

Příkaz **factor**:

Rozklad v Q[x]

```
> factor(Q);
```

$$(x^2-2)(x^2+1)$$

Rozklad v R[x]

```
> factor(Q,real);
```

$$(x+1.414213562)(x-1.414213562)(x^2+1.)$$

Rozklad v C[x]

```
> factor(Q,complex);
```

$$(x+1.414213562)(x+1.\,I)(x-1.\,I)(x-1.414213562)$$

Příkaz **polytools[split]**:

```
> Rozklad_Q:=polytools[split](Q,x);
```

$$Rozklad\_Q := (x - \mathrm{RootOf}(\_Z^2-2))(x + \mathrm{RootOf}(\_Z^2+1))(x + \mathrm{RootOf}(\_Z^2-2))$$
$$(x - \mathrm{RootOf}(\_Z^2+1))$$

```
> convert(Rozklad_Q,radical);
```

$$(x-\sqrt{2})(x+I)(x+\sqrt{2})(x-I)$$

Graf:

```
> plot(Q,x=-2..2,y=-3..2);
```




```
>
>
>
```